appraisals based on the number and distribution of the observations and the residual error obtained in fitting them to the long arc.

The prediction errors for satellites 1965 89A and 1968 2A appear to be consistent with the error in the Baker-Nunn field-reduced observations. Because these satellites were tracked with a high priority, many Baker-Nunn observations were obtained. It appears reasonable to expect that high-priority tracking, increased accuracy of field-reduced Baker-Nunn data, and larger numbers of laser observations should reduce the prediction error to less than 1 arcminute in the future.

References

[1] F. L. Whipple and C. A. Lundquist, "Tracking by the Smithsonian Astrophysical Observatory," *Phil. Trans. Roy. Soc. London*, ser. A, vol. 262, pp. 14–25, July 1967.

[2] C. A. Lundquist and G. Veis, Eds., "Geodetic parameters for a 1966 Smithsonian Institution standard earth," Smithsonian Astrophysical Observatory, Cambridge, Mass., Spec. Rept. 200, 3 vols., December 1966.

[3] C. A. Lundquist, "Satellite altimetry and orbit determination," Smithsonian Astrophysical Observatory, Cambridge, Mass., Spec. Rept. 248, August 18, 1967.

[4] C. A. Lundquist and H. D. Friedman, Eds., "Scientific horizons from satellite tracking," Smithsonian Astrophysical Observatory, Cambridge, Mass., Spec. Rept. 236, December 30, 1966.

[5] R. Rocherolles, J. Robieux, and G. Courrier, "Télémétrie précise à grande distance à l'aide du laser," *Onde Electrique*, no. 44, pp. 361–372, April 1964.

[6] G. D. Thayer, Environmental Science Services Administration, Boulder, Colo., private communication, October 23, 1967.

[7] B. O'Neil, "A comparison of satellite range measurements taken by the GSFC experimental and mobile laser systems," Wolf Research and Development Corp., Riverdale, Md., November 1968.

[8] C. G. Lehr, L. A. Maestre, and P. H. Anderson, "Satellite ranging with a laser and a correction for atmospheric refraction," *Österreich. Z. Vermess.*, spec. vol. 25, pp. 163–171, 1968.

[9] H. H. Plotkin, "The S-66 laser satellite tracking experiment," in *Quantum Electronics, III*, vol. 2, P. Grivet and N. Bloembergen, Eds. New York: Columbia University Press, 1964, pp. 1319–1332.

[10] T. S. Morrison and S. Ackerman, "Analysis of a programmed multipulse laser range measurement system," *Appl. Opt.*, vol. 6, pp. 1725–1727, October 1967.

Absolute and Differential Temperature Monitors Developed for Apollo Space Experiments

EARL E. SWARTZLANDER, JR., STUDENT MEMBER, IEEE

Abstract—**Investigation of the key parameters of temperature sensors suggests either platinum wire resistance sensors or thermistors. Thermistors have higher sensitivity and generally adequate stability, but have been scorned due to their nonlinear characteristics. A simple monitor is developed to linearize these nonlinear characteristics. A FORTRAN IV computer program is shown which optimizes all circuit values. The general technique should be applicable to other instrumentation problems.**

Introduction

THE SCIENTIFIC instruments which comprise the Apollo Telescope Mount require temperature monitors as part of the "housekeeping" requirements. The reason for this may be understood by realizing that these instruments are basically very high resolution telescopes. As an example, the ultraviolet scanning spectroheliometer (ATM Experiment S 055 A) which is being built for the Harvard College Observatory, Cambridge, Mass., requires a pointing accuracy of ± 5 seconds of arc [1]. With an instrument that requires pointing to these accuracy levels, care mut be taken to monitor the temperature of many points along the case. In the event of a failure of the thermal control system, such a network of temperature monitors can make possible correction of the data, as the bending and defocusing of the instrument can be calculated. The worth of such a monitoring scheme is contingent upon the availability of low cost, high reliability temperature monitors.

A brief survey indicated that commercial temperature monitors were not available which would meet the requirements. The prime requirements for the two basic monitors are as follows.

Absolute Monitor:

Temperature Range	60 to 90°F
Voltage Output	0 to 5 volts
Accuracy	±1°F

Differential Monitor:

Temperature Range	−5 to +5°F
Common Mode Range	60 to 90°F
Voltage Output	0 to 5 volts
Accuracy	±0.5°F

With the requirements thus defined, a brief summary of temperature sensor characteristics will be presented.

Manuscript received April 30, 1969. This paper was presented at the 1969 IEEE Internatl. Geoscience Electronics Symp., Washington, D. C., April 1969.

The author was with Ball Brothers Research Corporation, Boulder, Colo. 80302. He is now with Hughes Aircraft Company, Culver City, Calif. 90230.

TABLE I
TEMPERATURE SENSOR COMPARISON

Sensor	Sensitivity (per °C)	Stability (°C/yr)	Merit (yr/°C²)
Quartz crystal	0.0035 percent	±0.01	.35 percent
Platinum resistance	0.39 percent	±0.03	13 percent
Silicon bulk resistance	0.7 percent	±20	0.035 percent
$Pt\text{-}Pt_{90}RH_{10}$ thermocouple	0.004 percent	±1.5	0.003 percent
Thermistor	4 percent	±0.05	80 percent



Fig. 1. Comparison of resistance–temperature characteristics of thermistors and platinum wire resistance sensors.

SENSORS

Five basic types of temperature sensors were considered: linear temperature coefficient quartz crystals, platinum wire resistance sensors, silicon bulk resistance sensors (i.e., Sensistors[1]), thermocouples, and thermistors. The two most important characteristics of these sensors are sensitivity and long term stability. Table I compares these characteristics for each of the named sensors. It should be emphasized that the stability figures are for "off the shelf" commercial units. In most cases these can be improved by about an order of magnitude, if selection or special processing is used.

In addition, Table I shows a figure of merit for each of the sensors. The figure of merit is calculated using

$$\text{Merit} = \frac{\text{Sensitivity}}{\text{Stability}}. \quad (1)$$

As may be seen from Table I, the two sensors with the highest figure of merit are the platinum wire resistance sensor and the thermistor.

[1] Sensistor is a registered trademark of Texas Instruments, Inc.



Fig. 2. Photograph of Fenwal type GB42SMM1 subminiature thermistor.

Fig. 1 compares the resistance–temperature characteristics for the platinum wire resistance sensor (normalized to 10 kΩ at 0°C) with the characteristics of the thermistor. The thermistor is much more sensitive, but also quite nonlinear (i.e., $R_{\text{th}} \approx \alpha\epsilon^{-\beta t}$) in comparison to the platinum resistance sensor. The high sensitivity means that the amplifier used to realize the monitor operates with low closed loop gain and high stability.

Thus the decision was made to use thermistors as temperature sensors for the Apollo monitors, contingent upon the development of an electronic system which will linearize the nonlinear characteristics of the thermistor. The thermistor selected was Fenwal Electronics, Inc. type GB42SMM1 Oceanographic Iso-Curve[2] thermistor. A photograph of this device is shown on Fig. 2. The scale at the bottom of the photograph has 1/32 inch graduations; thus the thermistor is about 9/32 inch long exclusive of leads. As Fig. 2 indicates the active elements of the thermistor are two beads of sintered metallic oxides about 1/32 inch in diameter, sealed in small glass tubes. The sensor has very low thermal mass and a dissipation constant of 1.6 mW/°C in fast moving air [2].

ABSOLUTE TEMPERATURE MONITOR ELECTRONICS

A typical temperature monitor using a thermistor as the sensor is shown on Fig. 3. The output of the monitor is given by the equation

$$E_{\text{out}} = \frac{-V \cdot R_f}{R_{\text{in}} + R_{\text{eq}}} \cdot \frac{R_{\text{eq}}}{R_1} \quad (2)$$

where

$$R_{\text{eq}} = \frac{R_1 \cdot R_{\text{th}}}{R_1 + R_{\text{th}}}. \quad (3)$$

Conventional systems generally require that $R_1 \gg R_{\text{th}}$ and $R_{\text{in}} \gg R_{\text{th}}$. Thus (2) reduces to

$$E_{\text{out}} = \frac{-V \cdot R_f \cdot R_{\text{th}}}{R_{\text{in}} \cdot R_1}. \quad (4)$$

The relationship between E_{out} (from (4)) and temperature is essentially similar in shape to the thermistor's

[2] Iso-Curve is a registered trademark of Fenwal Electronics, Inc.



Fig. 3. Basic schematic diagram of absolute temperature monitor.



Fig. 5. Complete absolute temperature monitor schematic diagram showing all components.



Fig. 4. Schematic diagram of absolute temperature monitor showing the effect of source resistance on gain.

resistance–temperature curve. The conventional system thus has a high degree of nonlinearity.

Reducing R_{in} so that it is of the same order of magnitude as R_{eq} causes the first term of (2) to vary in such a manner as to compensate for the varying thermal sensitivity of the thermistor. Fig. 4 illustrates this concept. The gain of the closed loop operational amplifier is given by

$$A = \frac{-R_f}{R_{in} + R_{eq}} \tag{5}$$

where R_{eq} is the Thevenin equivalent source resistance of R_{th} and R_1. Thus when R_{th} is highest (and the sensitivity is greatest), the amplifier gain is lowest, and vice versa. Thus selection of the optimum value for R_{in} causes the gain of the amplifier to compensate for the nonlinear thermistor, yielding a linear output. This monitor retains the high sensitivity of the thermistor while generating a nearly linear output voltage.

Selection of the optimum value (for best linearity of the output voltage–temperature characteristics) of resistor R_{in} can be a difficult problem. The selection may be done by using an empirical trial and error procedure with a breadboard of the monitor, or by writing a computer program to select resistor values that will yield optimum results the first time a monitor is built. An advantage of the latter approach is that future temperature monitors can be "designed" by punching a data card and running the already developed computer program.

A computer program was written to select all of the resistor values for the monitor shown in Fig. 5. The program is written in ASA basic FORTRAN IV to provide maximum "portability" from one computer to another. A listing of the program is included in the Appendix. The values of resistors R_1 through R_5 (in Fig. 5) are selected to coincide with standard 1 percent resistor values, while bracketing values are computed for R_6 and R_7 which are selected during calibration at final assembly time to set the offset and gain of the individual monitor. The two adjustable resistors permit a ± 4 percent variation to make up for other component tolerances. An example of a typical computer printout is shown in Fig. 6. The particular printout is for the 60 to 90°F absolute temperature monitor for the Apollo Telescope Mount experiments. The nonlinearity is less than ± 0.050°F throughout the range, and is greatest at the extremes; for most readings the error due to nonlinearity is less than ± 0.025°F, which is less than the error due to telemetry quantization. The linearity improvement achieved with this monitor configuration is shown as a function of monitor range in Fig. 7. One reason for the reduced degree of improvement is that at higher temperatures the thermistor deviates some-

ABSOLUTE TEMPERATURE MONITOR

BRIDGE VOLTAGE 10.0 VOLTS
TEMPERATURE RANGE 60.0 TO 90.0 DEGREES F
OUTPUT RANGE 0.0 TO 5.0 VOLTS
AMPLIFIER BIAS CURRENT 200. NA.

R1 = 0.954E 05 OHMS PD = 0.159E-03 WATTS
R2 = 0.147E 06 OHMS
R3 = 0.261E 05 OHMS
R4 = 0.133E 05 OHMS
R5 = 0.154E 06 OHMS
R6 = 0.119E 04 OHMS NOM RANGE 0.274E 03 TO 0.243E 04 OHMS
R7 = 0.695E 04 OHMS NOM RANGE 0.162E 04 TO 0.143E 05 OHMS

TEMP. (F)	DESIRED OUTPUT	ACTUAL VOLTAGE	VOLTAGE DEVIATION	TEMP. DEVIATION
60.00000	0.00000	0.00793	0.00793	0.04760
61.00000	0.16666	0.17132	0.00465	0.02794
62.00000	0.33333	0.33528	0.00195	0.01170
63.00000	0.50000	0.49977	-0.00022	-0.00137
64.00000	0.66666	0.66474	-0.00192	-0.01152
65.00000	0.83333	0.83016	-0.00316	-0.01898
66.00000	1.00000	0.99599	-0.00400	-0.02401
67.00000	1.16666	1.16219	-0.00447	-0.02684
68.00000	1.33333	1.32871	-0.00462	-0.02773
69.00000	1.50000	1.49551	-0.00448	-0.02693
70.00000	1.66666	1.66255	-0.00411	-0.02467
71.00000	1.83333	1.82979	-0.00353	-0.02122
72.00000	2.00000	1.99719	-0.00280	-0.01682
73.00000	2.16666	2.16471	-0.00195	-0.01172
74.00000	2.33333	2.33230	-0.00102	-0.00616
75.00000	2.50000	2.49993	-0.00006	-0.00039
76.00000	2.66666	2.66755	0.00088	0.00533
77.00000	2.83333	2.83513	0.00179	0.01078
78.00000	3.00000	3.00261	0.00261	0.01570
79.00000	3.16666	3.16997	0.00331	0.01987
80.00000	3.33333	3.33717	0.00384	0.02305
81.00000	3.50000	3.50416	0.00416	0.02499
82.00000	3.66666	3.67091	0.00424	0.02548
83.00000	3.83333	3.83738	0.00404	0.02428
84.00000	4.00000	4.00352	0.00352	0.02116
85.00000	4.16666	4.16932	0.00265	0.01592
86.00000	4.33333	4.33472	0.00139	0.00834
87.00000	4.50000	4.49970	-0.00029	-0.00179
88.00000	4.66666	4.66421	-0.00244	-0.01469
89.00000	4.83333	4.82824	-0.00509	-0.03055
90.00000	5.00000	4.99173	-0.00826	-0.04956

Fig. 6. Results of the computer program which has optimized resistor values for the range of 60 to 90°F.

what from exponential characteristics which are so amenable to linearization by this method. Even so, the nonlinearity is less than 1 percent of the range for most of the cases shown on the graph.

Differential Temperature Monitor Electronics

The second type of monitor required for the Apollo experiments is a differential temperature monitor. The output voltage from such a monitor should be a linear function of the difference between two absolute temperatures. Fig. 8 shows one method to measure temperature differences. This method uses an operational amplifier to take the difference between two absolute monitor outputs. The main drawback with this scheme is that it uses a fairly large number of parts. The three operational amplifiers use more power (about 90 to 120 mW) than is desirable, so a circuit was developed to take the difference at the input to a single amplifier. This circuit is shown on Fig. 9. It should be noted that R_1, R_2, and R_{th1} make up a network similar to R_1, R_{in}, and R_{th} in Fig. 3; R_3, R_4, and R_{th2} make up a similar network. The currents i_1 and i_2 are the same nearly linear function of temperature as E_{out} for the absolute temperature monitor. Thus in Fig. 9,



Fig. 7. Graph comparing linearity of the temperature monitor output with the thermistor linearity.

$$i_i = \alpha_1 + \alpha_2 \cdot T_i \quad \text{for} \quad i = 1 \text{ or } 2. \tag{6}$$

Since $i_3 = i_1 - i_2$, the terms containing α_1 cancel leaving

$$i_3 = \alpha_2 \cdot (T_1 - T_2). \tag{7}$$

Analysis of the operational amplifier indicates that

$$E_{out} = R_5 \cdot \alpha_2 \cdot (T_1 - T_2) \tag{8}$$

which is the exact relationship desired for the differential temperature monitor. The circuit for the complete differential temperature monitor showing gain equalization, output offset, and differential gain adjusting resistors R_9, R_{10}, and R_{11}, respectively, is shown in Fig. 10. Note that this circuit uses less than a dozen precision resistors and a single operational amplifier to take high accuracy measurements of temperature differentials.

As was the case with the absolute temperature monitor, a computer program is used to determine the optimum values of the resistors which comprise the circuit. Resistors R_1 through R_8 are selected to be standard 1 percent values while the adjustment resistors R_9 through R_{11} are bracketed to less than a decade to ease the procurement task.

Monitor Performance

Accepting the thermistor manufacturer's specifications for long term stability of $\pm 0.05°C$ per year, the combined inaccuracy of the absolute temperature monitors is about $\pm 0.13°F$ [2]. This figure includes the worst case nonlinearity and sensor instability. Amplifier varia-



Fig. 8. Differential temperature monitor built with two absolute temperature monitors.



Fig. 9. Schematic of the basic differential temperature monitor which uses a single operational amplifier.

tions and component instability have a negligible effect on the output (as a result of low closed loop gain used). Laboratory tests indicate that the thermistor stability is generally better than about $\pm 0.02°$F per year. Using this figure results in an overall accuracy for the monitor of 0.07°F. The accuracy of the differential temperature monitors is about $\pm 0.1°$F if the thermistors are assumed to age in the same manner.

As with most space equipment, power consumption is a very important parameter. With these monitors the operational amplifier uses most of the power. For example, using a Fairchild Semiconductor type μA709A operational amplifier with ± 10 volt supplies, the



Fig. 10. Complete differential temperature monitor schematic diagram showing all components.

monitors each require about 30 to 40 mW [3]. If the operational amplifier was replaced with a National Semiconductor Corporation type NH0001 hybrid operational amplifier, the power consumption should drop to about 4 mW for the absolute monitor and about 5 mW for the differential monitor [4].

Summary

Temperature monitors have been developed using thermistors as the temperature sensors. Thermistors are a nearly ideal choice for the application, since they have high sensitivity and high stability, the only drawback is the need to linearize the exponential resistance–temperature characteristics. Circuits have been developed which monitor absolute as well as differential temperatures. These monitors are simple in concept. Indeed, to reduce the effort required to change the range, a computer program is used to optimize all resistor values.

The linearization technique should be applicable to many other resistive sensor instrumentation problems.

Appendix

The computer program and its three subroutines are listed below. The three subroutines RSTD, RTH, and SQFIT are listed after the listing of the main program.

The FORTRAN IV Program and Subroutines

```
      DIMENSION TH(42), TCOM(6), EO(6)
      COMMON RT(51), R1, V
1     READ (2,2) V,P,TMIN,TMAX,EMIN,EMAX,BIAS,I,IDEG,DSPAN
2     FORMAT (7F10.0,I1,A1,F8.0)
      IF (V) 4,3,4
3     STOP 7734
4     BIASI=BIAS*.000000001
      STEP=(TMAX-TMIN)*.02
      DO 5  J=1,51
      T=TMIN+STEP*FLOAT(J-1)
5     RT(J)=RTH(T,I)
      R=V*RT(1)
6     R1=R-(P*(RT(1)*(RT(1)+R+R)+R*R)-1000.*RT(1)*V*V)/((P+P)*(RT(1)+R))
      IF (ABS(R1-R)-R1/1000.) 8,7,7
7     R=R1
      GO TO 6
8     IF (R1-RT(1)) 9,10,10
9     R1=RT(1)
10    R1=RSTD(R1)
      P=V*V*RT(1)/((RT(1)+R1)*(RT(1)+R1))
      R=RSTD(RT(25))
      A=.39
11    RF=SQFIT(R)
      S=R*(1.+A)
      SF=SQFIT(S)
      IF (ABS(RF-SF)-ABS(RF*A)) 15,15,12
12    IF (SF-RF) 13,13,14
13    R=R*(1+.5*A)
      GO TO 11
14    R=R*(1-.5*A)
      GO TO 11
15    R=.5*(R+S)
      A=.5*A
      IF (A-.1) 16,16,11
16    R4=RSTD(R)
      IF (DSPAN) 37,17,17
17    WRITE (3,18) V,TMIN,TMAX,IDEG,EMIN,EMAX,BIAS
18    FORMAT (1H1,18X,28HABSOLUTE TEMPERATURE MONITOR//5X,14HBRIDGE VOLT
     1AGE,F5.1,6H VOLTS/5X,17HTEMPERATURE RANGE,F5.1,3H TO,F5.1,9H DEGRE
     2ES ,A1/5X,12HOUTPUT RANGE,F6.1,3H TO,F5.1,6H VOLTS/5X,22HAMPLIFIER
     3 BIAS CURRENT,F5.0,4H NA.//)
      R5=R1
      R3=R1
      RTEM=R1
      TEISO=RTH(TMIN-EMIN*(TMAX-TMIN)/(EMAX-EMIN),I)
      NCOU=-10
19    EQN22=R5*TEISO/(TEISO*(R1+R4+R5)+R1*(R4+R5))
      NCOU=NCOU+1
      R3OLD=R3
      EQN24=(R5*RT(26)*(R1+R4)+R1*R4*R5)/(RT(26)*(R1+R4+R5)+R1*(R4+R5))
      RTEMP=RTEM
      RTEM=R5
      R2=RSTD(EQN24/EQN22)
      R3=EQN24/(1.-EQN22)
      R5HAT=(EMAX-V*R3/(R2+R3))*(R1*R4+RT(51)*(R1+R4))/(BIASI*(R1*R4+RT(
     151)*(R1+R4))+V*((R1*R3-R2*RT(51))/(R2+R3)))
      IF (ABS(R5-R5HAT)-.01*R5HAT) 26,20,20
20    IF (NCOU) 22,22,21
21    R5=.5*(R5HAT+RTEM)
      R3=.5*(R3+R3OLD)
      GO TO 26
22    IF (R5+R5HAT) 23,23,24
23    R5=R5-R5HAT
      GO TO 19
24    R5=.5*(R5+R5HAT)
      IF (ABS(RTEMP-R5)-1000.) 19,25,19
25    R5=.5*(RTEM+RTEMP)
      GO TO 19
26    R6=R3
      R6LOW=RSTD(.01*R3)
      R6HI=RSTD(.09*R3)
      R3=RSTD(.95*R3)
      R6=R6-R3
      R7=R5
      R7LOW=RSTD(.01*R5)
      R7HI=RSTD(.09*R5)
      R5=RSTD(.95*R5)
      R7=R7-R5
27    EREF=(V-R2*BIASI)*(R3+R6)/(R2+R3+R6)
      IF (TMAX-TMIN-21.) 28,29,29
28    STEP=.05*(TMAX-TMIN)
      N=21
      GO TO 30
29    STEP=FLOAT(IFIX(TMAX-TMIN)/21)
      N=1+IFIX((TMAX-TMIN)/STEP)
30    A5=0.
      A6=0.
      A7=0.
      A8=0.
      A9=0.
      DO 31  J=1,N
      T=TMIN+STEP*FLOAT(J-1)
      TH(J)=RTH(T,I)
      EMON=EREF+(R5+R7)*(BIASI+(EREF*(TH(J)+R1)-TH(J)*V)/(R1*(TH(J)+R4)+
     1TH(J)*R4))
      A5=A5+1.
      A6=A6+T
      A7=A7+T*T
      A8=A8+EMON
31    A9=A9+T*EMON
      DENOM=A5*A7-A6*A6
      A1=(A7*A8-A6*A9)/DENOM
      A2=(A5*A9-A6*A8)/DENOM
      A21=(EMAX-EMIN)/(TMAX-TMIN)
      A11=EMIN-A21*TMIN
      R7=(R5+R7)*A21/A2
      R7LOW=RSTD(.01*R7)
      R7HI=RSTD(.09*R7)
      R5=RSTD(.95*R7)
      R7=R7-R5
      R6=(R3+R6)*(1.+(A11-A1)/100.)
      R6LOW=RSTD(.01*R6)
      R6HI=RSTD(.09*R6)
      R3=RSTD(.95*R6)
      R6=R6-R3
      IF (ABS(A11-A1)-.0002) 32,27,27
32    IF (ABS(A2-A21)-.0002) 33,27,27
33    WRITE (3,34) R1,P,R2,R3,R4,R5,R6,R6LOW,R6HI,R7,R7LOW,R7HI,IDEG
34    FORMAT (5X,4HR1 =,E10.3,5H OHMS,12X,5H PD =,E10.3,6H WATTS/5X,4HR2
     1 =,E10.3,5H OHMS/5X,4HR3 =,E10.3,5H OHMS/5X,4HR4 =,E10.3,5H OHMS/5
     2X,4HR5 =,E10.3,5H OHMS/5X,4HR6 =,E10.3,15H OHMS NOM RANGE,E10.3,3H
     3 TO,E10.3,5H OHMS/5X,4HR7 =,E10.3,15H OHMS NOM RANGE,E10.3,3H TO,E
     410.3,5H OHMS//5X,7HTEMP. (,A1,1H),5X,7HDESIRED,5X,6HACTUAL,5X,7HVO
     5LTAGE,6X,5HTEMP./21X,14HOUTPUT VOLTAGE,12X,9HDEVIATION)
      DO 35  J=1,N
      T=TMIN+STEP*FLOAT(J-1)
      ECALC=EMIN+(EMAX-EMIN)*STEP*FLOAT(J-1)/(TMAX-TMIN)
      EMON=EREF+(R5+R7)*(BIASI+(EREF*(TH(J)+R1)-TH(J)*V)/(R1*(TH(J)+R4)+
     1TH(J)*R4))
      DELTA=EMON-ECALC
      ERTEM=DELTA*(TMAX-TMIN)/(EMAX-EMIN)
35    WRITE (3,36) T,ECALC,EMON,DELTA,ERTEM
36    FORMAT (1H ,5F12.5)
      IF (DSPAN) 37,1,38
37    DSPAN=-DSPAN
38    DSPAM=-DSPAN
      WRITE (3,39) V,DSPAM,DSPAN,IDEG,TMIN,TMAX,IDEG,EMIN,EMAX,BIAS
39    FORMAT (33H1DIFFERENTIAL TEMPERATURE MONITOR//5X,14HBRIDGE VOLTAGE
     1F5.1,6H VOLTS/5X,18HDIFFERENTIAL RANGE,F5.1,3H TO,F5.1,9H DEGREES
     2,A1/5X,17HCOMMON MODE RANGE,F5.1,3H TO,F5.1,9H DEGREES ,A1/5X,12HO
     3UTPUT RANGE,F6.1,3H TO,F5.1,6H VOLTS/5X,22HAMPLIFIER BIAS CURRENT,
     4F5.0,4H NA.//)
      DR1=R1
      DR2=.95*DR1
      DR9LO=RSTD(.01*DR2)
      DR9HI=RSTD(.09*DR2)
      DR2=RSTD(DR2)
      DR9=DR1-DR2
      DR3=R4
      DR4=DR3
      RHO=RTH(.5*(TMAX+TMIN+DSPAN),I)
      CURA=RHO*V*(RHO+DR3)/(((DR1+DR3)*RHO+DR1*DR3)*DR3)
      RLO=RTH(.5*(TMAX+TMIN+DSPAM),I)
      CURB=RLO*V*(RLO+DR3)/(((DR1+DR3)*RLO+DR1*DR3)*DR3)
      DRF=2.5*(EMAX-EMIN)/(CURB-CURA)
      DR8=RSTD(.95*DRF)
      DR11L=RSTD(.01*DRF)
      DR11H=RSTD(.09*DRF)
      DR11=DRF-DR8
      DR5=DR1
      DR10=V*DR5*DRF/(V*DRF+.5*DR5*(EMAX+EMIN))
      DR6=RSTD(.95*DR10)
      DR10L=RSTD(.01*DR10)
      DR10H=RSTD(.09*DR10)
      DR10=DR10-DR6
      DR7=(RT(26)*DR1/(RT(26)+DR1)+DR3)*.5
      DR7=DR5*DR7/(DR5+DR7)
      DR7=DR7*(DR6+DR10)/(DR6+DR7+DR10)
      DR7=DRF*DR7/(DR7+DRF)
      WRITE (3,40) DR1,DR2,DR3,DR4,DR5,DR6,DR7,DR8,DR9,DR9LO,DR9HI,DR10,
     1DR10L,DR10H,DR11,DR11L
40    FORMAT (5X,5HDR1 =,E10.3,5H OHMS/5X,5HDR2 =,E10.3,5H OHMS/5X,5HDR3
     1 =,E10.3,5H OHMS/5X,5HDR4 =,E10.3,5H OHMS/5X,5HDR5 =,E10.3,5H OHMS
     2/5X,5HDR6 =,E10.3,5H OHMS/5X,5HDR7 =,E10.3,5H OHMS/5X,5HDR8 =,E10.
     33,5H OHMS/5X,5HDR9 =,E10.3,9H OHMS NOM,9X,5HRANGE,E10.3,3H TO,E10.
     43,5H OHMS/5X,6HDR10 =,E10.3,9H OHMS NOM,8X,5HRANGE,E10.3,3H TO,E10
     5.3,5H OHMS/5X,6HDR11 =,E10.3,9H OHMS NOM,8X,5HRANGE,E10.3,3H TO)
      DO 41  J=1,6
41    TCOM(J)=TMIN+.2*FLOAT(J-1)*(TMAX-TMIN)
      WRITE (3,42) DR11H, TCOM
42    FORMAT (1H+,55X,E10.3,5H OHMS//17H COMMON MODE TEMP,3X,6F10.4,10X,
     112HVOLTAGE DEV./18H DIFFERENTIAL TEMP)
      VREFR=-BIASI*DR7
      CUR3=(V-VREFR)/DR5-(V+VREFR)/(DR6+DR10)
      DO 44  K=1,21
      DEV=0.
      TEMPB=.05*DSPAN*FLOAT(K-11)
      DO 43  J=1,6
      RT1=RTH(TCOM(J)+TEMPB,I)
      RT2=RTH(TCOM(J)-TEMPB,I)
      CUR1=(V*RT1-VREFR*(DR1+RT1))/(RT1*(DR1+DR3)+DR1*DR3)
      CUR2=-(V*RT2+VREFR*(DR1+RT2))/(RT2*(DR1+DR3)+DR1*DR3)
      EO(J)=(BIASI-CUR1-CUR2-CUR3)*DRF+VREFR
43    DEV=DEV+(EO(J)-EMIN-.05*FLOAT(K-1)*(EMAX-EMIN))*(EO(J)-EMIN-.05*FL
     1OAT(K-1)*(EMAX-EMIN))
      TEMP1=FLOAT(K-11)*.1*DSPAN
      DEV=SQRT(.1666666667*DEV)
44    WRITE (3,45) TEMP1,EO,DEV
45    FORMAT (F10.1,10X,6F10.5,10X,F10.6)
      GO TO 1
      END
```

```
      FUNCTION RSTD (X)
      T=X
      I=0
      IF (X) 9,9,1
1     IF (T-100.) 2,3,3
2     T=10.*T
      I=I-1
      GO TO 1
3     IF (T-1000.) 5,4,4
4     T=.1*T
      I=I+1
      GO TO 3
5     RSTD=100.
6     IF  (ABS(T-RSTD)-.01214*RSTD) 8,8,7
7     RSTD=RSTD*1.02428
      GO TO 6
8     RSTD=FLOAT(IFIX(RSTD+.5))*10.**I
9     RETURN
      END
```

```
      FUNCTION RTH (T,I)
      S=T
      IF (I) 1,2,1
1     S=.5555555555*S-17.77777777
2     RTH=22420.*EXP(.8115-.0541*S)-(((((-.00001250757057*S+.00326802595
     1)*S-.3744326363)*S+23.75555491)*S-786.1372458)*S+7554.602361)
      RETURN
      END
```

```
      FUNCTION SQFIT (R)
      DIMENSION B(50)
      COMMON RT(51), R1, V
      S=0.
      SS=0.
      SY=0.
      SQFIT=0.
      DO 1  I=1,50
      B(I)=-V*R1*RT(I)/(RT(I)*(R1+R)+R1*R)
      S=S+B(I)
      SS=SS+B(I)*B(I)
1     SY=SY+B(I)*FLOAT(I)
      A1=(1275.*SS-S*SY)/(50.*SS-S*S)
      A2=(50.*SY-1275.*S)/(50.*SS-S*S)
      DO 2  I=1,50
2     SQFIT=(FLOAT(I)-A1-A2*B(I))*(FLOAT(I)-A1-A2*B(I))+SQFIT
      SQFIT=SQRT(SQFIT*.02)
      RETURN
      END
```

The criteria for stopping the various iterative loops are rather complex and will not be listed here [5].

Data input is by a single data card for each monitor. The data format is as follows.

Columns	Quantity	Units	Symbol
1–10	Bridge voltage	Volts	V
11–20	Maximum thermistor power dissipation	mW	PD
21–30	Minimum temperature	degrees*	TMIN
31–40	Maximum temperature	degrees*	TMAX
41–50	Minimum voltage output	volts	EMIN
51–60	Maximum voltage output	volts	EMAX
61–70	Operational amplifier bias current	nA	IBIAS
71	If input temperatures * are in °F = 1		
	If input temperatures * are in °C = 0		
72	If input temperatures * are in °F = "F"		
	If input temperatures * are in °C = "C"		
73–80	Differential temperature range	degrees*	DSPAN

If DSPAN < 0 only a differential monitor will be "designed"; if DSPAN = 0 only an absolute monitor will be "designed"; and if DSPAN > 0 both types will be "designed." For the differential monitors TMIN and TMAX define the extremes of the common mode range, and DSPAN is the differential range to be covered.

References

[1] "Design and performance specification, UV scanning polychromator spectroheliometer, ATM experiment S-055A," Ball Brothers Research Corporation, Boulder, Colo., Doc. 29540, 1968.

[2] "Oceanographic iso-curve sub-mini-probe thermistors," Fenwal Electronics, Inc., Framingham, Mass., Data Sheet D-10, 1966.

[3] "μA709A high performance operational amplifier," Fairchild Semiconductor, Mountain View, Calif., 1967.

[4] "NH0001 operational amplifier," National Semiconductor Corp., Santa Clara, Calif., 1968.

[5] E. E. Swartzlander, Jr., "Development of absolute and differential temperature monitors," M.S.E.E. thesis, University of Colorado, Boulder, 1969.